

10

Blending of Input Materials

10.1 Introduction

In a blending problem, there are:

- 1) Two or more input raw material commodities;
- 2) One or more qualities associated with each input commodity;
- 3) One or more output products to be produced by blending the input commodities, so certain output quality requirements are satisfied.

Blending models are used most frequently in three industries:

- 1) Feed and food (e.g., the blending of cattle feed, hotdogs, etc.);
- 2) Metals industry (e.g., the blending of specialty steels and nonferrous alloys, especially where recycled or scrap materials are used);
- 3) Petroleum industry (e.g., the blending of gasolines of specified octanes and volatility).

The market price of a typical raw material commodity may change significantly over the period of a month or even a week. The smart buyer will want to buy corn, for example, from the cheapest supplier. The even smarter buyer will want to exploit the fact that, as the price of corn drops relative to soybeans, the buyer may be able to save some money by switching to a blend that uses more corn.

A first approximation is that the quality of the finished product is the weighted average of the qualities of the products going into the blend. A listing of blending applications according important quality measures and typical input ingredients is given below:

Output Commodity	Qualities	Raw Materials
Feed	Moisture, density, fraction foreign material, fraction damaged.	Various types of feeds, e.g., by source.
Food	Protein, carbohydrate, fat content.	Corn, oats, soybeans, meat types.
Gasoline	Octane, volatility, vapor pressure.	Types of crude oil refinery products.
Metals	Carbon, manganese, chrome content.	Metal ore, scrap metals.
Grain for export	%Moisture, %foreign matter, %damaged.	Grain from various suppliers.
Coal for sale	%Sulfur, %BTU, %ash, %moisture.	Coal from Illinois, Wyoming, Pennsylvania.
Wine	Vintage, variety, region.	Pure wines of various vintage, variety, region.
Concrete	%CaO, %SiO ₂ , %Al ₂ O ₃ , %Fe ₂ O ₃ , %MgO, Strength, permeability to water, Cure time, workability, freeze resistance.	Portland cement, Slag, Fly ash, sand, stone/rocks of various size, water.
Natural gas	Heat content, Density.	Methane, Ethane, Propane, Nitrogen, Carbon dioxide.
Bank balance sheet	Proportion of loans of various types, average duration of loans and investment portfolios.	Types of loans and investments available.

Fields and McGee (1978) describe a feed blending LP for constructing low cost rations for cattle in a feedlot. Feedlot managers used this particular model at the rate of over 1,000 times per month. Schuster and Allen (1998) discuss the blending of grape juice at Welch's, Inc. The qualities of concern in grape juice are sweetness, acidity, and color. A blending problem must be solved at least once each season based upon how much of each type of grape is harvested by Welch's suppliers. Long term contracts require Welch's to take all of each supplier's harvest.

A recent success story in the steel industry has been the mini-mill. These small mills use mostly recyclable scrap steels to be charged into an electric furnace. The blending problem, in this case, is to decide what combination of scrap types to use to satisfy output quality requirements for specified products such as reinforcing bars, etc.

The first general LP to appear in print was a blending or diet problem formulated by George Stigler (1945). The problem was to construct a "recipe" from about 80 foods, so the mix satisfied about a dozen nutritional requirements. For example, percent protein greater than 5 percent, percent cellulose less than 40 percent, etc. When Stigler formulated this problem, the Simplex method for solving LPs did not exist. Therefore, it was not widely realized that this "diet problem" was just a special case of this wider class of problems. Stigler, realizing its generality, stated: "...there does not appear to be any direct method of finding the minimum of a linear function subject to linear conditions." The solution he obtained to his

specific problem by ingenious arguments was within a few cents of the least cost solution determined later when the Simplex method was invented. Both the least cost solution and Stigler's solution were not exactly haute cuisine. Both consisted largely of cabbage, flour and dried navy beans with a touch of spinach for excitement. It is not clear that anyone would want to exist on this diet or even live with someone who was on it. These solutions illustrate the importance of explicitly including constraints that are so obvious they can be forgotten. In this case, they are palatability constraints.

10.2 The Structure of Blending Problems

Let us consider a simple feed blending problem. We must produce a batch of cattle feed having a protein content of at least 15%. Mixing corn (which is 6% protein) and soybean meal (which is 35% protein) produces this feed.

In words, the protein constraint is:

$$\frac{\text{bushels of protein in mix}}{\text{bushels in mix}} \geq 0.15$$

If C is the number of bushels of corn in the mix and S is the number of bushels of soybean meal, then we have:

$$\frac{0.06 C + 0.35 S}{C + S} \geq 0.15$$

At first glance, it looks like we have trouble. This constraint is not linear. If, however, we multiply both sides by $C + S$, we get:

$$0.06 C + 0.35 S \geq 0.15 (C + S)$$

or, in standard form, finally:

$$-0.09 C + 0.20 S \geq 0.$$

Constraints on additional characteristics (i.e., fat, carbohydrates and even such slightly nonlinear things as color, taste, and texture) can be handled in similar fashion.

The distinctive feature of a blending problem is that the crucial constraints, when written in intuitive form, are *ratios* of linear expressions. They can be converted to linear form by multiplying through by the denominator. Ratio constraints may also be found in "balance sheet" financial planning models where a financial institution may have ratio constraints on the types of loans it makes or on the average duration of its investments.

The formulation is slightly more complicated if the blending aspect is just a small portion of a larger problem in which the batch size is a decision variable. The second example in this section will consider this complication. The first example will consider the situation where the batch size is specified beforehand.

10.2.1 Example: The Pittsburgh Steel Company Blending Problem

The Pittsburgh Steel (PS) Co. has been contracted to produce a new type of very high carbon steel which has the following tight quality requirements:

	At Least	Not More Than
Carbon Content	3.00%	3.50%
Chrome Content	0.30%	0.45%
Manganese Content	1.35%	1.65%
Silicon Content	2.70%	3.00%

PS has the following materials available for mixing up a batch:

	Cost per Pound	Percent Carbon	Percent Chrome	Percent Manganese	Percent Silicon	Amount Available
Pig Iron 1	0.0300	4.0	0.0	0.9	2.25	unlimited
Pig Iron 2	0.0645	0.0	10.0	4.5	15.00	unlimited
Ferro- Silicon 1	0.0650	0.0	0.0	0.0	45.00	unlimited
Ferro- Silicon 2	0.0610	0.0	0.0	0.0	42.00	unlimited
Alloy 1	0.1000	0.0	0.0	60.0	18.00	unlimited
Alloy 2	0.1300	0.0	20.0	9.0	30.00	unlimited
Alloy 3	0.1190	0.0	8.0	33.0	25.00	unlimited
Carbide (Silicon)	0.0800	15.0	0.0	0.0	30.00	20 lb.
Steel 1	0.0210	0.4	0.0	0.9	0.00	200 lb.
Steel 2	0.0200	0.1	0.0	0.3	0.00	200 lb.
Steel 3	0.0195	0.1	0.0	0.3	0.00	200 lb.

An one-ton (2000-lb.) batch must be blended, which satisfies the quality requirements stated earlier. The problem now is what amounts of each of the eleven materials should be blended together to minimize the cost, but satisfy the quality requirements. An experienced steel man claims the least cost mix will not use any more than nine of the eleven available raw materials. What is a good blend? Most of the eleven prices and four quality control requirements are negotiable. Which prices and requirements are worth negotiating?

Note the chemical content of a blend is simply the weighted average of the chemical content of its components. Thus, for example, if we make a blend of 40% Alloy 1 and 60% Alloy 2, the manganese content is $(0.40) \times 60 + (0.60) \times 9 = 29.4$.

10.2.2 Formulation and Solution of the Pittsburgh Steel Blending Problem

The PS blending problem can be formulated as an LP with 11 variables and 13 constraints. The 11 variables correspond to the 11 raw materials from which we can choose. Four constraints are from the upper usage limits on silicon carbide and steels. Four of the constraints are from the lower quality limits. Another four constraints are from the upper quality limits. The thirteenth constraint is the requirement that the weight of all materials used must sum to 2000 pounds.

If we let P_1 be the number of pounds of Pig Iron 1 to be used and use similar notation for the remaining materials, the problem of minimizing the cost per ton can be stated as:

```

MIN =      0.03 * P1 + 0.0645 * P2 + 0.065 * F1 + 0.061 * F2 + 0.1 * A1
+ 0.13 * A2 + 0.119 * A3 + 0.08 * CB + 0.021 * S1 + 0.02 * S2 + 0.0195
* S3;
! Raw material availabilities;
CB <= 20;
S1 <= 200;
S2 <= 200;
S3 <= 200;
! Quality requirements on;
! Carbon content;
.04 * P1 + 0.15 * CB + 0.004 * S1 + 0.001 * S2 + 0.001 * S3 >= 60;
.04 * P1 + 0.15 * CB + 0.004 * S1 + 0.001 * S2 + 0.001 * S3 <= 70;
! Chrome content;
0.1 * P2 + 0.2 * A2 + 0.08 * A3 >= 6;
0.1 * P2 + 0.2 * A2 + 0.08 * A3 <= 9;
! Manganese content;
0.009 * P1 + 0.045 * P2 + 0.6 * A1 + 0.09 * A2 + 0.33 * A3 + 0.009 *
S1 + 0.003 * S2 + 0.003 * S3 >= 27;
0.009 * P1 + 0.045 * P2 + 0.6 * A1 + 0.09 * A2 + 0.33 * A3 + 0.009 *
S1 + 0.003 * S2 + 0.003 * S3 <= 33;
! Silicon content;
0.0225 * P1 + 0.15 * P2 + 0.45 * F1 + 0.42 * F2 + 0.18 * A1 + 0.3 * A2
+ 0.25 * A3 + 0.3 * CB >= 54;
0.0225 * P1 + 0.15 * P2 + 0.45 * F1 + 0.42 * F2 + 0.18 * A1 + 0.3 * A2
+ 0.25 * A3 + 0.3 * CB <= 60;
! Finish good requirements;
P1 + P2 + F1 + F2 + A1 + A2 + A3 + CB + S1 + S2 + S3 = 2000;

```

In words, the general form of this model is:

Minimize cost of raw materials
subject to
(a) Raw material availabilities (rows 2-5)
(b) Quality requirements (rows 6-13)
(c) Finish good requirements (row 14)

It is generally good practice to be consistent and group constraints in this fashion.

For this particular example, when writing the quality constraints, we have exploited the knowledge that the batch size is 2000. For example, 3% of 2000 is 60, 3.5% of 2000 is 70, etc.

When solved, we get the solution:

Optimal solution found at step:		11
Objective value:		59.55629
Variable	Value	Reduced Cost
P1	1474.264	0.0000000
P2	60.00000	0.0000000
F1	0.0000000	0.1035937E-02
F2	22.06205	0.0000000
A1	14.23886	0.0000000
A2	0.0000000	0.2050311E-01
A3	0.0000000	0.1992597E-01
CB	0.0000000	0.3356920E-02
S1	200.0000	0.0000000
S2	29.43496	0.0000000
S3	200.0000	0.0000000
Row	Slack or Surplus	Dual Price
1	59.55629	1.0000000
2	20.00000	0.0000000
3	0.0000000	0.1771118E-03
4	170.5650	0.0000000
5	0.0000000	0.5000000E-03
6	0.0000000	-0.1833289
7	10.00000	0.0000000
8	0.0000000	-0.2547314
9	3.000000	0.0000000
10	0.0000000	-0.1045208
11	6.000000	0.0000000
12	0.0000000	-0.9880212E-01
13	6.000000	0.0000000
14	0.0000000	-0.1950311E-01

Notice only 7 of the 11 raw materials were used.

In actual practice, this type of LP was solved on a twice-monthly basis by Pittsburgh Steel. The purchasing agent used the first solution, including the reduced cost and dual prices, as a guide in buying materials. The second solution later in the month was mainly for the metallurgist's benefit in making up a blend from the raw materials actually on hand.

Suppose we can pump oxygen into the furnace. This oxygen combines completely with carbon to produce the gas CO_2 , which escapes. The oxygen will burn off carbon at the rate of 12 pounds of carbon burned off for each 32 pounds of oxygen. Oxygen costs two cents a pound. If you reformulated the problem to include this additional option, would it change the decisions? The oxygen injection option to burn off carbon is clearly uninteresting because, in the current solution, it is the lower bound constraint rather than the upper bound on carbon that is binding. Thus, burning off carbon by itself, even if it could be done at no expense, would increase the total cost of the solution.

10.3 A Blending Problem within a Product Mix Problem

One additional aspect of blending problem formulation will be illustrated with an example in which the batch size is a decision variable. In the previous example, the batch size was specified. In the following example, the amount of product to be blended depends upon how cheaply the product can be blended. Thus, it appears the blending decision and the batch size decision must be made simultaneously.

This example is suggestive of gasoline blending problems faced in a petroleum refinery. We wish to blend gasoline from three ingredients: butane, heavy naphtha, and catalytic reformat. Four characteristics of the resultant gasoline and its inputs are important: cost, octane number, vapor pressure, and volatility. These characteristics are summarized in the following table:

Feature	Commodity			Regular Gasoline (REG)	Premium Gasoline (PRM)
	Butane (BUT)	Catalytic Reformat (CAT)	Heavy Naphtha (NAP)		
Cost/Unit	7.3	18.2	12.5	-18.4	-22
Octane	120.0	100.0	74.0	$89 \leq \text{oct} \leq 110$	$94 \leq \text{oct} \leq 110$
Vapor Pressure	60.0	2.6	4.1	$8 \leq \text{vp} \leq 11$	$8 \leq \text{vp} \leq 11$
Volatility	105.0	3.0	12.0	$17 \leq \text{vo} \leq 25$	$17 \leq \text{vo} \leq 25$
Availability	1000.0	4000.0	5000.0	$4000 \leq \text{sell} \leq 8000$	$2000 \leq \text{sell} \leq 6000$

The cost per unit for *REG* and *PRM* are listed as negative, meaning we can sell them. That is, a negative cost is a revenue.

The octane rating is a measure of the gasoline's resistance to "knocking" or "pinging". Vapor pressure and volatility are closely related. Vapor pressure is a measure of susceptibility to stalling, particularly on an unusually warm spring day. Volatility is a measure of how easily the engine starts in cold weather.

From the table, we see in this planning period, for example, there are only 1,000 units of butane available. The profit contribution of regular gasoline is \$18.40 per unit *exclusive* of the cost of its ingredients.

A slight simplification assumed in this example is that the interaction between ingredients is linear. For example, if a "fifty/fifty" mixture of *BUT* and *CAT* is made, then its octane will be $0.5 \times 120 + 0.5 \times 100 = 110$ and its volatility will be $0.5 \times 105 + 0.5 \times 3 = 54$. In reality, this linearity is violated slightly, especially with regard to octane rating.

10.3.1 Formulation

The quality constraints require a bit of thought. The fractions of a batch of *REG* gasoline consisting of Butane, Catalytic Reformat, and Heavy Naphtha are *BUT/REG*, *CAT/REG*, and *NAP/REG*, respectively. Thus, if the god of linearity smiles upon us, the octane constraint of the blend for *REG* should be the expression:

$$(BUT/REG) \times 120 + (CAT/REG) \times 100 + (NAP/REG) \times 74 \geq 89.$$

Your expression, however, may be a frown because a ratio of variables like *BUT/REG* is definitely not linear. Multiplying through by *REG*, however, produces the linear constraint:

$$120 BUT + 100 CAT + 74 NAP \geq 89 REG$$

or in standard form:

$$120 BUT + 100 CAT + 74 NAP - 89 REG \geq 0.$$

10.3.2 Representing Two-sided Quality Constraints

All the quality requirements are two sided. That is, they have both an upper limit and a lower limit. The upper limit constraint on octane is clearly:

$$120 \text{ BUT} + 100 \text{ CAT} + 74 \text{ NAP} - 110 \text{ REG} \leq 0.$$

We can write it in equality form by adding an explicit slack:

$$120 \text{ BUT} + 100 \text{ CAT} + 74 \text{ NAP} - 110 \text{ REG} + \text{SOCT} = 0.$$

When $\text{SOCT} = 0$, the upper limit is binding. You can verify that, when $\text{SOCT} = 110 \text{ REG} - 89 \text{ REG} = 21 \text{ REG}$, the lower limit is binding. Thus, a compact way of writing both the upper and lower limits is with the two constraints:

- 1) $120 \text{ BUT} + 100 \text{ CAT} + 74 \text{ NAP} - 110 \text{ REG} + \text{SOCT} = 0$,
- 2) $\text{SOCT} \leq 21 \text{ REG}$.

Notice, even though there may be many ingredients, the second constraint involves only two variables. This is a compact way of representing two-sided constraints.

Similar arguments can be used to develop the vapor and volatility constraints. Finally, a constraint must be appended, which states the whole equals the sum of its raw material parts, specifically:

$$\text{REG} = \text{BUT} + \text{NAP} + \text{CAT}.$$

When all constraints are converted to standard form and the expression for profit contribution is written, we obtain the formulation:

```

MODEL:
MAX = 22 * B_PRM + 18.4 * B_REG - 7.3 * XBUT_PRM - 7.3 * XBUT_REG
      - 12.5 * XNAP_PRM - 12.5 * XNAP_REG
      - 18.2 * XCAT_PRM - 18.2 * XCAT_REG;
  ! Subject to raw material availabilities;
[RMLIMBUT] XBUT_PRM + XBUT_REG <= 1000;
[RMLIMCAT] XCAT_PRM + XCAT_REG <= 4000;
[RMLIMNAP] XNAP_PRM + XNAP_REG <= 5000;
!For each finished good, batch size computation;
[BDEF_REG] B_REG - XNAP_REG - XCAT_REG - XBUT_REG=0;
[BDEF_PRM] B_PRM - XNAP_PRM - XCAT_PRM - XBUT_PRM=0;
  ! Batch size limits;
[BLO_REG] B_REG >= 4000;
[BHI_REG] B_REG <= 8000;
[BLO_PRM] B_PRM >= 2000;
[BHI_PRM] B_PRM <= 6000;
  ! Upper(UP) and Lower(DN) quality restrictions for each product;
[QUPREGOC] - 110 * B_REG
           + SOCT_REG + 74 * XNAP_REG + 100 * XCAT_REG + 120 * XBUT_REG = 0;
[QDNREGOC] - 21 * B_REG + SOCT_REG <= 0;
[QUPREGVA] - 11 * B_REG
           + SVAP_REG + 4.1 * XNAP_REG + 2.6 * XCAT_REG + 60 * XBUT_REG = 0;
[QDNREGVA] - 3 * B_REG + SVAP_REG <= 0;
[QUPREGVO] - 25 * B_REG
           + SVOL_REG + 12 * XNAP_REG + 3 * XCAT_REG + 105 * XBUT_REG = 0;
[QDNREGVO] - 8 * B_REG + SVOL_REG <= 0;

```

```

[QUPPRMOC] - 110 * B_PRM
      + SOCT_PRM + 74 * XNAP_PRM + 100 * XCAT_PRM + 120 * XBUT_PRM = 0;
[QDNPRMOC] - 16 * B_PRM + SOCT_PRM <= 0;
[QUPPRMVA] - 11 * B_PRM
      + SVAP_PRM + 4.1 * XNAP_PRM + 2.6 * XCAT_PRM + 60 * XBUT_PRM = 0;
[QDNPRMVA] - 3 * B_PRM + SVAP_PRM <= 0;
[QUPPRMVO] - 25 * B_PRM
      + SVOL_PRM + 12 * XNAP_PRM + 3 * XCAT_PRM + 105 * XBUT_PRM = 0;
[QDNPRMVO] - 8 * B_PRM + SVOL_PRM <= 0;
END

```

The following is the same problem, set in a general, set-based blending formulation:

```

MODEL:
  ! General Blending Model (BLEND) in LINGO;
SETS:
  ! Each raw material has availability & cost/unit;
  RM/ BUT, CAT, NAP/: A, C;
  ! Each f. g. has min & max sellable, profit
  contr./unit and batch size to be determined;
  FG/ REG, PRM/: D, E, P, B;
  ! There are a set of quality measures;
  QM/ OCT, VAP, VOL/;
  ! Each RM & QM combo has a quality level;
  RQ( RM, QM): Q;
  ! For each combo QM, FG there are upper &
  lower limits on quality, slack on quality
  to be determined;
  QF( QM, FG): U, L, S;
  ! Each combination of RM and FG has an amount
  used, to be determined;
  RF( RM, FG): X;
ENDSETS
DATA:
  A= 1000, 4000, 5000; ! Raw material availabilities;
  C = 7.3, 18.2, 12.5;   ! R. M. costs;
  Q = 120, 60, 105, ! Quality parameters...;
      100, 2.6, 3,   ! R. M. by quality;
      74, 4.1, 12;
  D = 4000, 2000; ! Min needed of each F.G.;
  E = 8000, 6000; ! Max sellable of each F.G.;
  P = 18.4, 22;   ! Selling price of each F.G.;
  U = 110, 110, ! Upper limits on quality;
      11, 11, ! Quality by F.G.;
      25, 25;
  L = 89, 94, ! Lower limits on quality...;
      8, 8,   ! Quality by F.G.;
      17, 17;
ENDDATA

```

```

!-----;
! The model;
! For each raw material, the availabilities;
  @FOR( RM( I) :
    [RMLIM] @SUM( FG( K) : X( I, K) ) < A( I) ;
  );
  @FOR( FG( K) :
!For each finished good, compute batch size;
  [BDEF] B( K) = @SUM( RM( I) : X( I, K) );
  ! Batch size limits;
    [BLO] B( K) > D( K);
    [BHI] B( K) < E( K);
  ! Quality restrictions for each quality;
  @FOR( QM( J) :
[QUP]@SUM( RM(I) : Q(I, J) * X(I, K) ) + S( J,
  K) = U( J, K) * B( K);
[QDN] S(J, K) < (U(J, K) - L(J, K)) * B(K);
  ); );
!We want to maximize profit contribution;
[PROFIT] MAX = @SUM( FG: P * B)
  - @SUM( RM( I) : C( I) * @SUM( FG( K) : X( I, K) ));
END

```

As with all of our set based models, the data are well separated from the model equations. Thus, when the data change, the user need not be concerned with the model equations when updating the model.

The interesting part of the solution is:

Objective value:		48750.00
Variable	Value	Reduced Cost
B(REG)	4000.000	0.0000000
B(PRM)	4500.000	0.0000000
S(OCT, REG)	84000.00	0.0000000
S(OCT, PRM)	72000.00	0.0000000
S(VAP, REG)	1350.424	0.0000000
S(VAP, PRM)	7399.576	0.0000000
S(VOL, REG)	17500.00	0.0000000
S(VOL, PRM)	36000.00	0.0000000
X(BUT, REG)	507.4153	0.0000000
X(BUT, PRM)	492.5847	0.0000000
X(CAT, REG)	1409.958	0.0000000
X(CAT, PRM)	2590.042	0.0000000
X(NAP, REG)	2082.627	0.0000000
X(NAP, PRM)	1417.373	0.0000000
Row	Slack or Surplus	Dual Price
RMLIM(BUT)	0.0000000	27.05000
RMLIM(CAT)	0.0000000	6.650000
RMLIM(NAP)	1500.000	0.0000000
BDEF(REG)	0.0000000	-22.65000
BLO(REG)	0.0000000	-1.225000
BHI(REG)	4000.000	0.0000000
QUP(REG, OCT)	0.0000000	-0.4750000
QDN(REG, OCT)	0.0000000	0.4750000

QUP (REG, VAP)	0.0000000	0.0000000
QDN (REG, VAP)	10649.58	0.0000000
QUP (REG, VOL)	0.0000000	0.0000000
QDN (REG, VOL)	14500.00	0.0000000
BDEF (PRM)	0.0000000	-22.65000
BLO (PRM)	2500.000	0.0000000
BHI (PRM)	1500.000	0.0000000
QUP (PRM, OCT)	0.0000000	-0.4750000
QDN (PRM, OCT)	0.0000000	0.4750000
QUP (PRM, VAP)	0.0000000	0.0000000
QDN (PRM, VAP)	6100.424	0.0000000
QUP (PRM, VOL)	0.0000000	0.0000000
QDN (PRM, VOL)	0.0000000	0.0000000
PROFIT	48750.00	1.0000000

The solution suggests that Premium is the more profitable product, so we sell the minimum amount of Regular required and then sell as much Premium as scarce resources, BUT and CAT, allow.

LP blending models have been a standard operating tool in refineries for years. Recently, there have been some instances where these LP models have been replaced by more sophisticated nonlinear models, which more accurately approximate the nonlinearities in the blending process. See Rigby, Lasdon, and Waren (1995), for a discussion of how Texaco does it. For example, volatility may be represented by a logarithmic expression and octane may be represented with a polynomial like $a_1*x + a_2*x^2 + a_3*x^3 + a_4*x^4$, see Rardin(1998).

There is a variety of complications as gasoline blending models are made more detailed. For example, in high quality gasoline, the vendor may want the octane to be constant across volatility ranges in the ingredients. The reason is, if you “floor” the accelerator on a non-fuel injected automobile, a shot of raw gas is squirted into the intake. The highly volatile components of the blend will reach the combustion chamber first. If these components have low octane, you will have knocking, even though the “average” octane rating of the gasoline is high. This may be more important in a station selling gas for city driving than in a station on a cross country highway in Kansas where most driving is at a constant speed.

10.3.3 Representing Soft Target Quality Constraints

Recall that in the above gasoline blending problem we required the octane rating of Regular gasoline to be in the range: $89 \leq \text{octane} \leq 110$. Now suppose that ideally, the octane target is 91. If we required the octane of Regular to be exactly 91, we could write:

$$74 * \text{XNAP_REG} + 100 * \text{XCAT_REG} + 120 * \text{XBUT_REG} = 91 * \text{B_REG};$$

But if we allow deviations above and below, we could add total or absolute Lower and Upper deviation/slack/surplus variables, TLOCT_REG and TUOCT_REG :

$$\begin{aligned} 74 * \text{XNAP_REG} + 100 * \text{XCAT_REG} + 120 * \text{XBUT_REG} \\ + \text{TLOCT_REG} - \text{TUOCT_REG} &= 91 * \text{B_REG}; \quad ! \text{ Allow deviations from 91;} \\ \text{TLOCT_REG} &\leq (91-89) * \text{B_REG}; \quad ! \text{ But not lower than 89;} \\ \text{TUOCT_REG} &\leq (110-91) * \text{B_REG}; \quad ! \text{ But not higher than 110;} \end{aligned}$$

In the (maximize) objective we could add terms:

```
- ALPHA * SLOCT_REG - ALPHAU * SUOCT_REG;
```

Setting constants ALPHA, ALPHAU > 0 would encourage the blended octane to be close to 91. Setting ALPHA, ALPHAU = 0 would only restrict octane to the interval [89, 110].

An alternative way of requiring the octane to be exactly 91 would be with the constraint:

```
( 74 * XNAP_REG + 100 * XCAT_REG + 120 * XBUT_REG) / B_REG = 91;
```

If we allow deviations above and below, we could add relative Lower and Upper deviation/slack/surplus variables, RLOCT_REG and RUOCT_REG:

```
( 74 * XNAP_REG + 100 * XCAT_REG + 120 * XBUT_REG) / B_REG
+ RLOCT_REG - RUOCT_REG = 91; ! Allow deviations from 91;
```

A computational disadvantage of this form is that it is nonlinear, even if we multiply by B_REG.

10.3.4 Discrete Blending/All-or-Nothing Usage

There are some blending applications in which it makes sense to use only discrete quantities of certain ingredients. For example, if you are doing menu planning for a school or some other institution, you would not recommend that a meal consist of 0.75 apples and 1.2 bananas. You would recommend only a whole number. In the processing of scrap metal, the scrap tends come in compressed bundles. If you are melting scrap in a furnace, you would use all or nothing of a bundle, not a fraction. If you are blending coal at a port facility, you would tend to use either all or none of a barge of coal, or a railcar of coal. This all-or-nothing feature can be handled straightforwardly by introducing binary variables. In our previous gasoline blending example, suppose there is a naptha supply of 4100, and it is of an all or nothing nature, i.e., if you use any of it you must use all of it. This can be represented by introducing a binary variable: ZNAP = 1 if any (and all) or the naptha is used, = 0 if none is used. The additional constraints would be:

```
@BIN( ZNAP); ! ZNAP is a 0/1, binary variable.;
[RMLIMNAP] XNAP_PRM + XNAP_REG = 4100 * ZNAP; ! Naptha used = 0 or 4100;
```

When the model is solved with this constraint replacing:

```
[RMLIMNAP] XNAP_PRM + XNAP_REG = 5000;
```

we get a solution which in part is:

```
Global optimal solution found.
Objective value:    45810.00
```

Variable	Value
B_PRM	2700.000
B_REG	6400.000
XBUT_PRM	295.5508
XBUT_REG	704.4492
XNAP_PRM	850.4237
XNAP_REG	3249.576

XCAT_PFM	1554.025
XCAT_REG	2445.975
ZNAP	1.000000

10.3.5 Treatments vs. Ingredients in Blending Problems

In some blending problems, you can apply treatments that can significantly change the quality directly without having much effect on the volume of the blend. Examples are: a) in gasoline blending, the octane rating can be increased by adding small amounts of tetraethyl lead (TEL), methyl tertiary-butyl ether (MTBE), or ferrocene. Note, however, that regulatory authorities in many countries may have restrictions on the use of these additives because of pollution issues. b) in the manufacture of steel, one can remove C (carbon) and S (sulfur) by treating with O (oxygen) and CaO (lime).

A simplified representation of (b) is that C can be removed by injecting O to produce CO, which is easily separated out. Atomic weights of C and O are 12 and 16. So each ton of O injected might remove up to $12/16 = 0.75$ tons of C. Another simplified representation is that S can be removed by adding CaO, which produces CaS + O, both of which are easily separated out, e.g. as slag floating on the top of the blending vessel. Atomic weights of Ca and S are 40 and 32. So each ton of CaO applied might remove up to $32/(40+16) = 0.57$ tons of S.

We can represent treatments as a generalization of raw material ingredients if we introduce an additional parameter vector:

$vc(i)$ = volume contribution per unit of treatment i applied.

We also use the notation:

Parameters:

$q(i, j)$ = contribution of treatment i to quality j .

$U(j)$ = upper limit on quality j per volume, e.g., the fraction of sulfur allowed,

Variables:

B = batch size, to be determined,

$x(i)$ = amount of treatment or raw material i to be used,

For a typical ingredient, $vc(i) = 1$, e.g., if we add a ton of Wyoming coal to a batch of blends of various coals, the batch size increases by 1 ton. Now consider in contrast that we add a ton of lime to a steel batch. The purpose of the lime is, among other things, to remove sulfur. The lime combines with the sulfur which is then removed as slag from the top of the blending vessel. Let us suppose that each ton of lime removes 0.57 tons of sulfur. In this case, we would set $vc(\text{lime}) = -0.57$, i.e., adding lime leads to a net decrease in the final batch size. Further, $q(\text{lime}, \text{sulfur}) = -0.57$, i.e., each ton of lime added, decreases the amount of sulfur in the final batch by 0.57 tons.

The constraints in general are then:

$B = \sum_i vc(i) * x(i)$; Compute the batch size;

For each quality j :

$(\sum_i q(i, j) * x(i)) / B \leq U(j)$; Quality constraint in ratio form.

or in linear form:

$\sum_i q(i, j) * x(i) \leq U(j) * B$;

10.4 Choice of Alternate Interpretations of Quality Requirements

Some quality features can be stated according to some measure of either goodness or, alternatively, undesirability. An example is the efficiency of an automobile. It could be stated in miles per gallon or alternatively in gallons per mile. In considering the quality of a blend of ingredients (e.g., the efficiency of a fleet of cars), it is important to identify whether it is the goodness or the badness measure which is additive over the components of the blend. The next example illustrates.

A federal regulation required the average of the miles per gallon computed over all automobiles sold by an automobile company in a specific year be at least 18 miles per gallon.

Let us consider a hypothetical case for the Ford Motor Company. Assume Ford sold only the four car types: Mark V, Ford, Granada, and Fiesta. Various parameters of these cars are listed below:

Car	Miles per Gallon	Marginal Prod. Cost	Selling Price
Fiesta	30	13,500	14,000
Granada	18	14,100	15,700
Ford	16	14,500	15,300
Mark V	14	15,700	20,000

There is some flexibility in the production facilities, so capacities may apply to pairs of car types. These limitations are:

Yearly Capacity in Units	Car Types Limited
250,000	Fiestas
2,000,000	Granadas plus Fords
1,500,000	Fords plus Mark V's

There is a sale capacity limit of 3,000,000 on the total of all cars sold. How many of each car type should Ford plan to sell?

Interpreting the mileage constraint literally results in the following formulation:

$$\begin{aligned}
 \text{MAX} &= 500 * \text{FIESTA} + 1600 * \text{GRANADA} + 4300 * \text{MARKV} + 800 * \text{FORD}; \\
 12 * \text{FIESTA} & & & - 4 * \text{MARKV} - 2 * \text{FORD} & \geq 0; \\
 \text{FIESTA} & & & & & \leq 250; \\
 & & \text{GRANADA} & & & + \text{FORD} \leq 2000; \\
 & & & & \text{MARKV} & + \text{FORD} \leq 1500; \\
 \text{FIESTA} & + \text{GRANADA} & + \text{MARKV} & + \text{FORD} & \leq 3000;
 \end{aligned}$$

Automobiles and dollars are measured in 1000s. Note row 2 is equivalent to:

$$\frac{30 \text{ Fiesta} + 18 \text{ Granada} + 16 \text{ Ford} + 14 \text{ Mark V}}{\text{Fiesta} + \text{Granada} + \text{Ford} + \text{Mark V}} \geq 18.$$

The solution is:

Optimal solution found at step:		1
Objective value:		6550000.
Variable	Value	Reduced Cost
FIESTA	250.0000	0.0000000
GRANADA	2000.000	0.0000000
MARKV	750.0000	0.0000000
FORD	0.0000000	2950.000
Row	Slack or Surplus	Dual Price
1	6550000.	1.000000
2	0.0000000	-1075.000
3	0.0000000	13400.00
4	0.0000000	1600.000
5	750.0000	0.0000000
6	0.0000000	0.0000000

Let's look more closely at this solution. Suppose each car is driven the same number of miles per year regardless of type. An interesting question is whether the ratio of the total miles driven by the above fleet divided by the number of gallons of gasoline used is at least equal to 18. Without loss, suppose each car is driven one mile. The gasoline used by a car driven one mile is 1/(miles per gallon). Thus, if all the cars are driven the same distance, then the ratio of miles to gallons of fuel of the above fleet is $(250 + 2000 + 750)/[(250/30) + (2000/18) + (750/14)] = 17.3$ miles per gallon—which is considerably below the mpg we thought we were getting.

The first formulation is equivalent to allotting each automobile the same number of gallons and each automobile then being driven until it exhausts its allotment. Thus, the 18 mpg average is attained by having less efficient cars drive fewer miles. A more sensible way of phrasing things is in terms of gallons per mile. In this case, the mileage constraint is written:

$$\frac{\text{Fiesta}/30 + \text{Granada}/18 + \text{Ford}/16 + \text{MarkV}/14}{\text{Fiesta} + \text{Granada} + \text{Ford} + \text{MarkV}} \leq 1/18$$

Converted to standard form this becomes:

$$-0.022222222 * \text{FIESTA} + 0.0069444444 * \text{FORD} + 0.015873016 * \text{MARKV} = 0;$$

When this problem is solved with this constraint, we get the solution:

Optimal solution found at step:		0
Objective value:		4830000.
Variable	Value	Reduced Cost
FIESTA	250.0000	0.0000000
FORD	0.0000000	2681.250
MARKV	350.0000	0.0000000
GRANADA	2000.000	0.0000000

Notice the profit contribution drops noticeably under this second interpretation. The federal regulations could very easily be interpreted to be consistent with the first formulation. Automotive companies, however, wisely implemented the second way of computing fleet mileage rather than leave

themselves open to later criticism of having implemented what Uncle Sam said rather than what he meant.

For reference, in 2010, the U.S. "light truck" (so-called sport utility vehicles) fleet mileage requirement was 23.4 miles per gallon, and the passenger car fleet requirement was 27.5 miles per gallon. For each tenth of a mile per gallon that a fleet falls short of the requirement, the U.S. Federal government sets a fine of \$5 per vehicle. The requirements are based on a "model year" basis. This gives a car manufacturer some flexibility if it looks like it might miss the target in a given year. For example, the manufacturer could "stop production" of a vehicle that has poor mileage, such as the big Chevy Suburban, and declare that all subsequent copies sold belong to the next model year. This may achieve the target in the current model year, but postpone the problem to the next model year.

10.5 How to Compute Blended Quality

The general conclusion is one should think carefully when one needs to compute an average performance measure for some blend or collection of things. There are several ways of computing averages or means when one has a collection of N quantities, x_1, x_2, \dots, x_N :

Type	Formula	Average of (5, 9)
Arithmetic:	$(x_1 + x_2 + \dots + x_N)/N$,	7.000
Logarithmic:	$(x_1 - x_2)/\text{LN}(x_1/x_2)$, for $N=2$	6.805
Geometric:	$(x_1 \times x_2 \dots \times x_N)^{(1/N)}$	6.708
Harmonic:	$1/[(1/x_1 + 1/x_2 + \dots + 1/x_N)/N]$	6.429

The arithmetic mean is appropriate for computing the mean return of the assets in a portfolio. If, however, we are interested in the average growth of a portfolio over time, we would probably want to use the geometric mean of the yearly growths. Consider, for example, an investment that has a growth factor of 1.5 in the first year and 0.67 in the second year (e.g., a rate of return of 50% in the first year and -33% in the second year). Most people would *not* consider the average growth to be $(1.5 + 0.67)/2 = 1.085$.

The harmonic mean tends to be appropriate when computing an average rate of something, as in average miles/gallon in the example above, or for computing the average density of blend of ingredients. Density is usually measured in weight per volume (e.g., grams per cubic centimeter). If the decision variables are measured in weight units rather than volume units, then the harmonic mean is appropriate. The harmonic mean is also appropriate for computing the average price earnings ratio for a collection of companies.

The logarithmic mean is used in computing the average temperature difference in a heat exchanger, based on the temperature difference at the two ends, e.g., as used in a petroleum refinery.

10.5.1 Example

We have two ingredients, one with a density of 0.7 g/cc and the other with a density of 0.9 g/cc. If we mix together one gram of each, what is the density of the mix? Clearly, the mix has a weight of 2 grams. Its volume in cc's is $1/0.7 + 1/0.9$. Thus, its density is $2/(1/0.7 + 1/0.9) = 0.7875$ g/cc. This is less than the 0.8 we would predict if we took the arithmetic average. If we define:

$$X_i = \text{grams of ingredient } i \text{ in the mix,}$$

$$t = \text{target lower limit on density desired.}$$

Then, we can write the density constraint for our little example as:

$$\begin{aligned} (X_1 + X_2)/(X_1/0.7 + X_2/0.9) &\geq t, \\ \text{or} \\ (X_1 + X_2)/t &\geq X_1/0.7 + X_2/0.9, \\ \text{or} \\ (1/t - 1/0.7) X_1 + (1/t - 1/0.9) X_2 &\geq 0, \end{aligned}$$

(i.e., a harmonic mean constraint).

10.5.2 Generalized Mean

One can generalize the idea just discussed by introducing a transformation $f(q)$. The interpretation is that the function $f()$ “linearizes” the quality. The basic idea is that many of the quality measures used in practice were chosen somewhat arbitrarily (e.g., why is the freezing point of water 32 degrees on the Fahrenheit scale?). So, even though a standardly used quality measure does not “blend linearly”, perhaps we can find a transformation that does. Such linearizations are common in industry. Some examples follow:

1. The American Petroleum Institute likes to measure the lightness of a material in “API gravity”, see Dantzig and Thapa (1997). Water has an API gravity of 10. API gravity does not blend linearly. However, the specific gravity, defined by:

$$sg = 141.5/(API \text{ gravity} + 131.5)$$

does blend linearly. Note, the specific gravity of a material is the weight in grams of one cubic centimeter of material. For example, if component 1 has an API gravity of 35, component 2 has a API gravity of 55, x_i is the amount used of component i , and we want a blend with an API gravity of at most 45, the constraint could be written:

$$(141.5/(35 + 131.5)) \times x_1 + (141.5/(55 + 131.5)) \times x_2 \geq 141.5/(45 + 131.5) (x_1 + x_2).$$

Note, if we want the API gravity to be low, then we want the specific gravity high.

2. In the transmissivity of light through a glass fiber of length x_i , or the financial growth of an investment over a period of length x_i , or in the probability of no failures in a number of trials x_i , one may have constraints of the form: $a_1^{x_1} a_2^{x_2} \dots a_n^{x_n} \geq a_0$. This can be linearized by taking logarithms (e.g., $\ln(a_1) * x_1 + \ln(a_2) * x_2 + \dots \ln(a_n) * x_n \geq \ln(a_0)$).

For example, if we expect stocks to have a long term growth rate of 10% per year, we expect less risky bonds to have a long term growth rate of 6% per year, we want an overall growth of 40% over five years, and x_1 and x_2 are the number of years we invest in stocks and bonds respectively over a five year period, then we want the constraint:

$$(1.10)^{x_1} (1.06)^{x_2} \geq 1.40.$$

Linearizing, this becomes:

$$\begin{aligned} \ln(1.10) x_1 + \ln(1.06) x_2 &\geq \ln(1.40), \text{ or} \\ .09531 x_1 + .05827 x_2 &\geq .3364, \\ x_1 + x_2 &= 5. \end{aligned}$$

3. Rigby, Lasdon, and Waren (1995) use this idea when approximating the Reid vapor pressure (RVP) of a blended gasoline at Texaco. Note, the RVP of a liquid is the pressure in a closed container having a small amount of the liquid at 100 degrees F. If r_i is the RVP of component i of the blend, they use the transformation:

$$f(r_i) = r_i^{1.25}$$

For example, if component 1 has an RVP of 80, component 2 has an RVP of 100, x_i is the amount used of component i , and we want a blend with an RVP of at least 90, the constraint could be written:

$$80^{1.25} \times x_1 + 100^{1.25} \times x_2 \geq 90^{1.25} \times (x_1 + x_2),$$

or

$$239.26 \times x_1 + 316.23 x_2 \geq 277.21 (x_1 + x_2).$$

4. The *flashpoint* of a chemical is the lowest temperature at which it will catch fire. Typical jet fuel has a flashpoint of around 100 degrees F. Typical heating oil has a flashpoint of at least 130 degrees F. The jet fuel used in the supersonic SR-71 jet aircraft had a flashpoint of several hundred degrees F. If p_i is the flashpoint of component i in degrees F, then the transformation, the so-called blending index is:

$$f(p_i) = 51708 * \exp(((\text{LOG}(p_i) - 2.6287)^2 / (-0.91725)))$$

where LOG is the natural logarithm, will approximately linearize the flashpoint. See Fahim et al. (2010) or Kaiser et al. (2020). For example, if component 1 has a flashpoint of 100, and component 2 has a flashpoint of 140, then $f(100) = 731.073$, and $f(140) = 151.569$. Notice that $f(p_i)$ is a decreasing function of p_i , so a higher flashpoint means a lower blending index $f(p_i)$ value.

Suppose we want a blend with a flashpoint of at least 130. Now $f(130) = 218.939$, so the flash point constraint, where x_j is the amount by weight of component j of the above two components is:

$$731.073 \times x_1 + 151.569 \times x_2 \leq 218.939 \times (x_1 + x_2).$$

5. The *viscosity* of a liquid is a measure, in units of centistokes, of the time it takes a standard cup volume of liquid, at 122 degrees Fahrenheit, to flow through a hole of a certain diameter. The higher the viscosity, the less quickly the liquid flows. If v_i is the viscosity of component i , then the transformation:

$$f(v_i) = \ln(\ln(v_i + .08))$$

will approximately linearize the viscosity.

For example, if component 1 has a viscosity of 5, component 2 has a viscosity of 25, x_i is the amount used of component i , and we want a blend with a viscosity of at most 20, the constraint would be written:

$$\ln(\ln(5 + .08)) \times x_1 + \ln(\ln(25 + .08)) \times x_2 \leq$$

$$\ln(\ln(20 + .08)) \times (x_1 + x_2),$$

or

$$.4857 x_1 + 1.17 x_2 \leq 1.0985(x_1 + x_2).$$

The preceding examples apply the transformation to each quality individually. One could extend the idea even further by allowing a “matrix” transformation to several qualities together.

10.5.3 Blending Formula for Energy Density of a Gas

When you turn on your gas stove (or gas turbine if you are in industry) each morning and you set the dial to "5", you would like to get the same amount of heat from the burner from one day to the next, even though the composition of the "natural" gas feeding your appliance may vary from day to day. If the Wobbe index of the gas remains constant from day to day then the heat output of your appliance will remain the same from day to day. The Wobbe index is a measure of the heat content of the gas that flows through an orifice of constant size under constant pressure per unit of time. Based on elementary physics, Goffredo Wobbe in 1927 deduced that the heat content is a) proportional to the heat content of a cubic foot (or meter) of gas at standard temperature and pressure (STP), and b) inversely proportional to the square root of the density (or specific gravity) of the gas. This is because at STP, lighter gas flows through the orifice faster than heavier gas;

```
! Blending gases based on energy density, the Wobbe index. (WobbeIndex.lng);
SETS:
    GAS: BTUpCF, SPG, AVAIL, X;
ENDSETS
DATA:
    ! The gases of interest, their heat content in
      BTU per cubic foot at STP, and specific gravity
      relative to air, cubic feet available per unit time;
    GAS,      BTUpCF,  SPG,  AVAIL=
    ARGON      0      1.38  10
    HYDROGEN   325    0.07  10
    HELIUM     0      0.138 45
    METHANE    1012   0.55  120
    ETHANE     1773   1.04  235
    PROPANE    2522   1.52  827
    BUTANE     3225   2.01  395
    NatGas     1025   0.65  690
    CO2        0      1.52  150
    NITROGEN   0      0.97  80
    ;
! We want the Wobbe index in this interval;
    WOBL = 1500;  WOBV = 1600;
ENDDATA

SUBMODEL Wobbe:
! Variables:
    X( i) = cubic feet of gas i included in blend;
! Availability limits;
    @FOR( GAS(i):
        X(i) <= AVAIL(i);
    );
! batch size;
    BLENDTOT = @SUM( GAS(i): X( i));
! Total heat content;
    HEATTOT = @SUM( GAS(i): BTUpCF( i)* X( i));
```

```

! Total weight in SPG;
  SPGTOT = @SUM( GAS(i): SPG( i)* X( i));
! heat content per cf at STP (numerator of Wobbe index);
  HEATcf = HEATTOT/ BLENDTOT;    ! Nonlinear;
! SPG per cf at STP (prepare denominator);
  SPGcf = SPGTOT/ BLENDTOT;    ! Nonlinear;
! Wobbe index;
  WobbeI = HEATcf/( SPGcf^0.5); ! Nonlinear;
! bounds on the Wobbe index;
  WobbeI >= WOBL;
  WobbeI <= WOBU;
! A plausible objective is to maximize the energy content;
  MAX = HEATTOT;
ENDSUBMODEL

CALC:
! Set some parameters before the solve;
  @SET( 'TERSEO',1); ! Output(0:verbose, 1:terse, 2:only errors, 3:none);
  @SET( 'GLOBAL', 1);! 0:Do not use Global solver, 1:Use the Globasolver;
  @SOLVE( Wobbe); ! Solve a specified submodel;

@WRITE( 'Solution Status = ', ISTAT, @NEWLINE( 1));
! Display a very simple solution report;
@WRITE(@NEWLINE(1), ' A Simple Customized Solution Report', @NEWLINE( 1));
@WRITE( 'Objective = ', HEATTOT, @NEWLINE(1));
@WRITE( 'Wobbe index= ', WobbeI, @NEWLINE(1));
@WRITE( '      Gas      Amount to use ', @NEWLINE(1));
@FOR( GAS( i) | X( i) #GT# 0:
! Can format output if desired;
  @WRITE( @FORMAT( GAS( i),'10s'), ' ', @FORMAT( X, '12.3f'), @NEWLINE(1));
);
ENDCALC

```

The solution is:

Global optimal solution found.
Objective value: 4342384.

```

A Simple Customized Solution Report
Objective = 4342383.857923649
Wobbe index= 1600
      Gas      Amount to use
ARGON          10.000
HYDROGEN       10.000
HELIUM         45.000
METHANE        120.000
ETHANE         235.000
PROPANE        827.000
BUTANE         312.588
NATGAS         690.000
CO2            150.000

```

NITROGEN 80.000

Notice that all gases, except Butane, are used to their maximum availability.

10.6 Interpretation of Dual Prices for Blending Constraints

The dual price for a blending constraint usually requires a slight reinterpretation in order to be useful. As an example, consider the minimum octane constraint for Premium gasoline in the model considered earlier. The constraint was effectively:

$$-94 B_PRM + 120 XBUT_PRM + 74 XNAP_PRM + 100 XCAT_PRM \geq 0.$$

The dual price of this constraint is the rate of increase in profit if the right-hand side of this constraint is increased from 0 to 1. Unfortunately, this is not a change we would ordinarily consider. More typical changes that might be entertained would be changing the octane rating from 94 to either 93 or 95. A very approximate rule for estimating the effect of changing the coefficient in row i of variable B_PRM is to compute the product of the dual price in row i and the value of variable B_PRM . For variable B_PRM and the octane constraint, this value is $-.475 \times 4500 = -2137.5$. This suggests, if the octane requirement is reduced to 93 (or increased to 95) from 94, the total profit will increase by about 2137.5 to $48750 + 2137.5 = \$50887.5$ (or decrease to $48750 - 2137.5 = \$46,612.5$). If the LP is actually re-solved with an octane requirement of 93 (or 95), the actual profit contribution changes to \$51,000 (or \$46,714.29).

This approximation can be summarized generally as follows:

If we wish to change a certain quality requirement of blend by a small amount δ , the effect on profit of this change is approximately of the magnitude $\delta \times$ (dual price of the constraint) \times (batch size). For small changes, the approximation tends to understate profit after the change. For large changes, the approximation may err in either direction.

10.7 Fractional or Hyperbolic Programming

In blending problems, we have seen ratio constraints of the form:

$$\frac{\sum_i q_i X_i}{\sum_j X_j} \geq q_0$$

can be converted to linear form, by rewriting:

$$\sum_j q_j X_j \geq q_0 \sum X_j \quad \text{or} \quad \sum (q_j - q_0) x_j \geq 0$$

Can we handle a similar feature in the objective? That is, can a problem of the following form be converted to linear form?

$$(1) \text{ Maximize } \frac{u_o + \sum_j u_j X_j}{v_o + \sum_j v_j X_j}$$

$$(2) \text{ subject to: } \sum_j a_{ij} X_j = b_i, \text{ for } i = 1, 2, \dots$$

The a_{ij} , u_0 , u_j , v_0 , and v_j are given constants. For example, we might wish to maximize the fraction of protein in a blend subject to constraints on availability of materials and other quality specifications.

We can make it linear with the following transformations:

Define:

$$r = 1/(v_0 + \sum_j v_j X_j)$$

and

$$y_j = X_j r$$

We assume $r > 0$.

Then our objective is:

$$(1') \text{ Maximize } u_0 r + \sum_j u_j y_j$$

subject to:

$$r = 1/(v_0 + \sum_j v_j X_j), \text{ or}$$

$$(1.1') \quad r v_0 + \sum_j v_j y_j = 1$$

Any other constraint i of the form:

$$(2) \quad \sum_j a_{ij} X_j = b_i$$

can be written as:

$$\sum_j a_{ij} X_j r = b_i r$$

or linear in terms of the new variables,

$$(2') \quad \sum_j a_{ij} y_j - b_i r = 0$$

10.8 Multi-Level Blending: Pooling Problems


A complicating factor in some blending problems is that not all raw ingredients can be stored separately. Such a situation can arise in a number of ways. Two ingredients may be produced at the same location, but for economic reasons, are transported together (e.g., in one tank car or via one pipeline). Another possibility is two ingredients are delivered separately, but only a single holding facility is available at the blending site. In general, many facilities that blend ingredients have only a modest number of storage facilities. For example, a grain storage facility may have only a half dozen bins. A petroleum refinery may have only a half dozen tanks. If there are more than a half dozen different sources of raw materials, then not all raw materials can be stored separately. In the petroleum industry, this leads to what is called a pooling problem.

This pooling of raw materials within a blending problem leads to a nonlinear program. The pooling problem discussed here is taken from Haverly (1978). A , B , and C are ingredients containing 3%, 1%, and 2% sulfur as an impurity, respectively. These chemicals are to be blended to provide two output products, X and Y , which must meet sulfur content upper limits of 2.5% and 1.5%, respectively. At the given prices of \$9 per unit of X and \$15 per unit of Y , customers will buy all of X and Y produced up to a maximum of 100 units of X and 200 units of Y . The costs per unit for ingredients A , B , and C are \$6, \$16, and \$10, respectively. The problem is to operate the process in order to maximize profit.

A complicating factor in this blending process is the fact that products A and B must be stored in the same tank, or "pool". So, until the amounts of A and B are determined, the pool sulfur content is unknown. Figure 10.1 illustrates. However, it is the pool sulfur content together with the amounts of

pool material and of chemical C used in blending X and Y that determine the X and Y sulfur contents. The sulfur constraints on X and Y affect the amounts of A and B needed, and it is this “circularity” that causes a nonlinearity.

Figure 10.1 A Pooling Problem



The constraint equations defining this system involve material balances and sulfur constraints for the output products. Consider the material balance equations first.

We have the following mass balance for the pool, assuming all of the pool material is to be used up:

$$\text{Amount } A + \text{Amount } B = \text{Pool to } X + \text{Pool to } Y.$$

For the output products, the balance equations are:

$$\text{Pool to } X + C \text{ to } X = \text{Amount } X$$

and:

$$\text{Pool to } Y + C \text{ to } Y = \text{Amount } Y.$$

For the total amount of C , the equation is:

$$C \text{ to } X + C \text{ to } Y = \text{Amount } C.$$

Introducing the pool sulfur percent, $\text{Pool } S$, as a new variable, makes it easy to write the X and Y sulfur constraints. If we let $\text{Pool } S$ have a value between 0 and 100 and express all other percentages on the same scale, these constraints are:

$$\text{Pool } S \times \text{Pool to } X + 2 \times C \text{ to } X \leq 2.5 \times \text{Amount } X$$

$$\text{Pool } S \times \text{Pool to } Y + 2 \times C \text{ to } Y \leq 1.5 \times \text{Amount } Y$$

The left-hand side of each inequality represents the actual sulfur content of the appropriate product and the right-hand side is the maximum amount of sulfur permitted in that product. The pool sulfur balance equation is:

$$3 \times \text{Amount } A + 1 \text{ Amount } B = \text{Pool } S \times (\text{Amount } A + \text{Amount } B).$$

This defines Pool S as the amount of sulfur in the pool divided by the total amount of material in the pool.

As mentioned earlier, product demand sets upper bounds on production as:

$$\text{Amount } X \leq 100$$

$$\text{Amount } Y \leq 200$$

and physical considerations restrict all variables to be nonnegative quantities. Clearly, the pool sulfur can never be less than 1% or more than 3%. Thus:

$$1 \leq \text{Pool } S \leq 3.$$

Finally, the profit function must be formulated. If $\text{Cost}A$, $\text{Cost}B$, $\text{Cost}C$, $\text{Cost}X$, and $\text{Cost}Y$ are the appropriate cost coefficients, the profit can be written as:

$$\begin{aligned} & \text{Cost}X \times \text{Amount}X + \text{Cost}Y \times \text{Amount}Y - \text{Cost}A \times \text{Amount}A \\ & - \text{Cost}B \times \text{Amount}B - \text{Cost}C \times \text{Amount}C \end{aligned}$$

A LINGO formulation follows:

```
MODEL:
MAX = 9 * AMOUNTX + 15 * AMOUNTY
      - 6 * A2POOL - 16 * B2POOL - 10 * AMOUNTC;
! Sources = uses for the pool;
A2POOL + B2POOL = POOL2X + POOL2Y;
! Sources for final products;
POOL2X + C2X = AMOUNTX;
POOL2Y + C2Y = AMOUNTY;
! Uses of C;
AMOUNTC = C2X + C2Y;
! Demand upper limits;
AMOUNTX <= 100;
AMOUNTY <= 200;
! Blending constraint for sulfur in the pool;
3 * A2POOL + B2POOL = POOLS * ( A2POOL + B2POOL ); ! Nonlinear;
! Blending constraints for sulfur in final products;
POOLS * POOL2X + 2 * C2X <= 2.5 * AMOUNTX;      ! Nonlinear;
POOLS * POOL2Y + 2 * C2Y <= 1.5 * AMOUNTY;      ! Nonlinear;
END
```

This problem is tricky in that it has (as we shall see) several local optima. LINGO, using the default “local” solver, may find the following solution:

```

Local optimal solution found.
Objective value:      400.0000
Variable             Value             Reduced Cost
A2POOL              0.0000             2.000000
B2POOL             100.0000             0.000000
AMOUNTC            100.0000             0.000000
POOLS              1.0000             0.000000
POOL2X             0.0000             7.000000
POOL2Y            100.0000             0.000000
C2X                0.0000             1.000000
C2Y               100.0000             0.000000
AMOUNTX            0.0000             0.000000
AMOUNTY           200.0000             0.000000

```

Examination of the solution shows the optimal operation produces only product *Y* using equal amounts of *B* and *C*. The cost per unit of output is $\$(16 + 10) / 2 = \13 and the sale price is \$15, giving a profit of \$2 per unit. Since all 200 units are produced and sold, the profit is \$400.

Nonlinear problems, such as this pooling model, have the curious feature that the solution you get may depend upon where the solver starts its solution search. You can set the starting point by inserting an "INIT" initialization section at the beginning of your model such as the following:

```

INIT:
! A solution with obj = 100;
A2POOL = 50;
B2POOL = 0;
AMOUNTC = 50;
POOLS = 3;
POOL2X = 50;
POOL2Y = 0;
C2X = 50;
C2Y = 0;
AMOUNTX = 100;
AMOUNTY = 0;
ENDINIT

```

The INIT section allows you to provide the solver with an initial guess at the solution. Starting at this point provided in the INIT section, LINGO may find the solution:

```

Local optimal solution found.
Objective value:      100.0000

Variable             Value
A2POOL              50.0000
B2POOL              0.0000
AMOUNTC            50.0000
POOLS              3.0000
POOL2X             50.0000
POOL2Y              0.0000
C2X                50.0000
C2Y                0.0000

```

```

AMOUNTX      100.0000
AMOUNTY      0.0000

```

In this solution, only product X is produced and sold. It is made using an equal blend of chemicals A and C . The net cost of production is \$8 per unit, yielding a profit of \$1 per unit of X sold. Since only 100 units are called for, the final profit is \$100. This solution is locally optimal. That is, small changes from this operating point reduce the profit. There are no feasible operating conditions close to this one that yield a better solution.

If you specify an all zero solution in the INIT section, you will get the all zero solution with an objective of zero. There are in fact three local optima, one with objective 0, another with objective 100, and another with an objective of 400.

Our earlier solution, yielding a profit of \$400, is also a local optimum. However, there is no other feasible point with a larger profit, so we call the \$400 solution a global optimum.

This model is an example of where a global solver is helpful. If the “Global solver” option is selected in LINGO, then the global optimal solution with value 400 is found without fail. For this problem, all solutions obtained have the property that many constraints are active. In other words, they hold as equalities

The difficulty in solving a nonlinear problem often increases with the number of nonlinear constraints and nonlinear variables in the problem. The above formulation has three nonlinear constraints. In a more realistic problem involving pooling, there may be a half a dozen quality types, in addition to the one above involving sulfur. Because there are 3 nonlinear constraints involving sulfur, there would be $3 \times 6 = 18$ nonlinear constraints if there were 6 different quality features to consider. Can we find a different formulation for the pooling problem in which the number of nonlinear constraints does not increase so much with number of quality features? The following formulation shows that the answer is yes.

MODEL:

```

MAX = 9 * AMOUNTX + 15 * AMOUNTY
      - 6 * A2POOL - 16 * B2POOL - 10 * AMOUNTC;;
! Sources = uses for the pool;
A2POOL + B2POOL = POOL2X + POOL2Y;
! Sources for final products;
POOL2X + C2X = AMOUNTX;
POOL2Y + C2Y = AMOUNTY;
! Uses of C;
AMOUNTC = C2X + C2Y;! Demand upper limits;
AMOUNTX <= 100;
AMOUNTY <= 200;

! Path flow/ PQ formulation.
POFA, POFB = proportion of input to the pool from sources A and B;
POFA + POFB = 1;
! Path flows: Source -> Pool -> Destination;
PFLOAX = POFA* POOL2X; ! Nonlinear;
PFLOBX = POFB* POOL2X; ! Nonlinear;
PFLOAY = POFA* POOL2Y; ! Nonlinear;
PFLOBY = POFB* POOL2Y; ! Nonlinear;
! Link the path flows to the link flows;
A2POOL = PFLOAX + PFLOAY;
B2POOL = PFLOBX + PFLOBY;

```

```
POOL2X = PFLOAX + PFLOBX;
POOL2Y = PFLOAY + PFLOBY;
```

```
! Blending constraints for final products, sulfur delivered;
3* PFLOAX + 1* PFLOBX + 2 * C2X <= 2.5 * AMOUNTX;
3* PFLOAY + 1* PFLOBY + 2 * C2Y <= 1.5 * AMOUNTY;
```

```
END
```

This formulation might be called a path formulation because it introduces variables, $PFLOAX$, $PFLOAY$, $PFLOBX$, and $PFLOBY$. For example, $PFLOBX$ is the flow on the path from B through Pool to X. Variations of this formulation are also called a PQ formulation. See Gupte et al.(2017). It has four nonlinear constraints, however, they do not involve sulfur related variables. Thus, even if there are more than one quality features to be modeled, there will still be only 4 nonlinear constraints. Gupte et al. show that computationally, this formulation is much better on larger problems, especially when the Global solver is used.

10.9 Problems

- The Exxoff Company must decide upon the blends to be used for this week's gasoline production. Two gasoline products must be blended and their characteristics are listed below:

Gasoline	Vapor Pressure	Octane Number	Selling Price (in \$/barrel)
Lo-lead	≤ 7	≥ 80	\$ 9.80
Premium	≤ 6	≥ 100	\$12.00

The characteristics of the components from which the gasoline can be blended are shown below:

Component	Vapor Pressure	Octane Number	Available this Week (in barrels)
Cat-Cracked Gas	8	83	2700
Isopentane	20	109	1350
Straight Gas	4	74	4100

The vapor pressure and octane number of a blend is simply the weighted average of the corresponding characteristics of its components. Components not used can be sold to "independents" for \$9 per barrel.

- What are the decision variables?
- Give the LP formulation.
- How much Premium should be blended?

2. The Blendex Oil Company blends a regular and a premium product from two ingredients, Heptane and Octane. Each liter of regular is composed of exactly 50% Heptane and 50% Octane. Each liter of premium is composed of exactly 40% Heptane and 60% Octane. During this planning period, there are exactly 200,000 liters of Heptane and 310,000 liters of Octane available. The profit contributions per liter of the regular and premium product this period are \$0.03 and \$0.04 per liter respectively.
- Formulate the problem of determining the amounts of the regular and premium products to produce as an LP.
 - Determine the optimal amounts to produce without the use of a computer.
3. Hackensack Blended Whiskey Company imports three grades of whiskey: Prime, Choice, and Premium. These unblended grades can be used to make up the following two brands of whiskey with associated characteristics:

Brand	Specifications	Selling price per liter
Scottish Club	Not less than 60% Prime. Not more than 20% Premium.	\$6.80
Johnny Gold	Not more than 60% Premium. Not less than 15% Prime.	\$5.70

The costs and availabilities of the three raw whiskeys are:

Whiskey	Available This Week (Number of Liters)	Cost per Liter
Prime	2,000	\$7.00
Choice	2,500	\$5.00
Premium	1,200	\$4.00

Hackensack wishes to maximize this week's profit contribution and feels it can use linear programming to do so. How much should be made of each of the two brands? How should the three raw whiskeys be blended into each of the two brands?

4. The Sebastopol Refinery processes two different kinds of crude oil, Venezuelan and Saudi, to produce two general classes of products, Light and Heavy. Either crude oil can be processed by either of two modes of processing, Short or Regular. The processing cost and amounts of Heavy and Light produced depend upon the mode of processing used and the type of crude oil used. Costs vary, both across crude oils and across processing modes. The relevant characteristics are summarized in the table below. For example, the short process converts each unit of Venezuelan crude to 0.45 units of Light product, 0.52 units of Heavy product, and 0.03 units of waste.

	Short Process		Regular Process	
	Venezuela	Saud	Venezuela	Saud
Light product fraction	0.45	0.60	0.49	0.68
Heavy product fraction	0.52	0.36	0.50	0.32
Unusable product fraction	0.03	0.04	0.01	0.00

Saudi crude costs \$20 per unit, whereas Venezuelan crude is only \$19 per unit. The short process costs \$2.50 per unit processed, while the regular process costs \$2.10 per unit. Sebastopol can process 10,000 units of crude per week at the regular rate. When the refinery is running the Short process for the full week, it can process 13,000 units per week.

The refinery may run any combination of short and regular processes in a given week.

The respective market values of Light and Heavy products are \$27 and \$25 per unit. Formulate the problem of deciding how much of which crudes to buy and which processes to run as an LP. What are the optimal purchasing and operating decisions?

5. There has been a lot of soul searching recently at your company, the Beansoul Coal Company (BCC). Some of its better coal mines have been exhausted and it is having more difficulty selling its coal from remaining mines. One of BCC's most important customers is the electrical utility, Power to the People Company (PPC). BCC sells coal from its best mine, the Becky mine, to PPC. The Becky mine is currently running at capacity, selling all its 5000 tons/day of output to PPC. Delivered to PPC, the Becky coal costs BCC \$81/ton and PPC pays BCC \$86/ton. BCC has four other mines, but you have been unable to get PPC to buy coal from these mines. PPC says that coal from these mines does not satisfy its quality requirements. Upon pressing PPC for details, it has agreed it would consider buying a mix of coal as long as it satisfies the following quality requirements: sulfur $\leq 0.6\%$; ash $\leq 5.9\%$; BTU ≥ 13000 per ton; and moisture $\leq 7\%$. You note your Becky mine satisfies this in that its quality according to the above four measures is: 0.57%, 5.56%, 13029 BTU, and 6.2%. Your four other mines have the following characteristics:

Mine	BTU Per Ton	Sulfur Percent	Ash Percent	Moisture Percent	Cost Per Ton Delivered to PPC
Lex	14,201	0.88	6.76	5.1	73
Casper	10,630	0.11	4.36	4.6	90
Donora	13,200	0.71	6.66	7.6	74
Rocky	11,990	0.39	4.41	4.5	89

The daily capacities of your Lex, Casper, Donora, and Rocky mines are 4000, 3500, 3000, and 7000 tons respectively. PPC uses an average of about 13,000 tons per day.

BCC's director of sales was ecstatic upon hearing of your conversation with PPC. His response was "Great! Now, we will be able sell PPC all of the 13,000 tons per day it needs". Your stock with BCC's newly appointed director of productivity is similarly high. Her reaction to your discussion with PCC was: "Let's see, right now we are making a profit contribution of only \$5/ton of coal sold to PPC. I have figured out we can make a profit contribution of \$7/ton if we can sell them a mix. Wow! You are an ingenious negotiator!" What do you recommend to BCC?

6. The McClendon Company manufactures two products, bird food and dog food. The company has two departments, blending and packaging. The requirements in each department for manufacturing a ton of either product are as follows:

	Time per Unit in Tons	
	Blending	Packaging
Bird food	0.25	0.10
Dog food	0.15	0.30

Each department has 8 hours available per day.

Dog food is made from the three ingredients: meat, fishmeal, and cereal. Bird food is made from the three ingredients: seeds, ground stones, and cereal. Descriptions of these five materials are as follows.

Descriptions of Materials in Percents					
	Protein	Carbohydrates	Trace Minerals	Abrasives	Cost (in \$/ton)
Meat	12	10	1	0	600
Fishmeal	20	8	2	2	900
Cereal	3	30	0	0	200
Seeds	10	10	2	1	700
Stones	0	0	3	100	100

The composition requirements of the two products are as follows:

Composition Requirements of the Products in Percents					
	Protein	Carbohydrates	Trace Minerals	Abrasive	Seeds
Bird food	5	18	1	2	10
Dog food	11	15	1	0	0

Bird food sells for \$750 per ton while dog food sells for \$980 per ton. What should be the composition of bird food and dog food and how much of each should be manufactured each day?

7. Recent federal regulations strongly encourage the assignment of students to schools in a city, so the racial composition of any school approximates the racial composition of the entire city. Consider the case of the Greenville city schools. The city can be considered as composed of five areas with the following characteristics:

Area	Fraction Minority	Number of students
1	0.20	1,200
2	0.10	900
3	0.85	1,700
4	0.60	2,000
5	0.90	2,500

The ruling handed down for Greenville is that a school can have neither more than 75 percent nor less than 30 percent minority enrollment. There are three schools in Greenville with the following capacities:

School	Capacity
Bond	3,900
Pocahontas	3,100
Pierron	2,100

The objective is to design an assignment of students to schools, so as to stay within the capacity of each school and satisfy the composition constraints while minimizing the distance traveled by students. The distances in kilometers between areas and schools are:

School	Area				
	1	2	3	4	5
Bond	2.7	1.4	2.4	1.1	0.5
Pocahontas	0.5	0.7	2.9	0.8	1.9
Pierron	1.6	2.0	0.1	1.3	2.2

There is an additional condition that no student can be transported more than 2.6 kilometers. Find the number of students that should be assigned to each school from each area. Assume any group of students from an area has the same ethnic mix as the whole area.

8. A farmer is raising pigs for market and wishes to determine the quantity of the available types of feed that should be given to each pig to meet certain nutritional requirements at minimum cost. The units of each type of basic nutritional ingredient contained in a pound of each feed type is given in the following table along with the daily nutritional requirement and feed costs.

Nutritional Ingredient	Pound of Corn	Pound of Tankage	Pound of Alfalfa	Units Required per day
Carbohydrates	9	2	4	20
Proteins	3	8	6	18
Vitamins	1	2	6	15
Cost (cents)/lb.	7	6	5	

9. Rico-AG is a German fertilizer company, which has just received a contract to supply 10,000 tons of 3-12-12 fertilizer. The guaranteed composition of this fertilizer is (by weight) at least 3% nitrogen, 12% phosphorous, and 12% potash. This fertilizer can be mixed from any combination of the raw materials described in the table below.

Raw Material	% Nitrogen	% Phosphorous	% Potash	Current World Price/Ton
AN	50	0	0	190 Dm
SP	1	40	5	180 Dm
CP	2	4	35	196 Dm
BG	1	15	17	215 Dm

Rico-AG has in stock 500 tons of SP that was bought earlier for 220 Dm/ton. Rico-AG has a long-term agreement with Fledermausguano, S.A. This agreement allows it to buy already mixed 3-12-12 at 195 Dm/ton.

- Formulate a model for Rico-AG that will allow it to decide how much to buy and how to mix. State what assumptions you make with regard to goods in inventory.
- Can you conclude in advance that no CP and BG will be used because they cost more than 195 Dm/ton?

10. The Albers Milling Company buys corn and wheat and then grinds and blends them into two final products, Fast-Gro and Quick-Gro. Fast-Gro is required to have at least 2.5% protein while Quick-Gro must have at least 3.2% protein. Corn contains 1.9% protein while wheat contains 3.8% protein. The firm can do the buying and blending at either the Albers (*A*) plant or the Bartelso (*B*) plant. The blended products must then be shipped to the firm's two warehouse outlets, one at Carlyle (*C*) and the other at Damiansville (*D*). Current costs per bushel at the two plants are:

	A	B
Corn	10.0	14.0
Wheat	12.0	11.0

Transportation costs per bushel between the plants and warehouses are:

		Fast-Gro:		Quik-Gro:	
		To		To	
		C	D	C	D
From	A	1.00	2.00	A	3.00 3.50
	B	3.00	0.75	B	4.00 1.90

The firm must satisfy the following demands in bushels at the warehouse outlets:

Warehouse	Product	
	Fast-Gro	Quik-Gro
C	1,000	3,000
D	4,000	6,000

Formulate an LP useful in determining the purchasing, blending, and shipping decisions.

11. A high quality wine is typically identified by three attributes: (a) its vintage, (b) its variety, and (c) its region. For example, the Optima Winery of Santa Rosa, California produced a wine with a label that stated: 1984, Cabernet Sauvignon, Sonoma County. The wine in the bottle may be a blend of wines, not all of which need be of the vintage, variety, and region specified on the label. In this case, the state of California and the U.S. Department of Alcohol, Tobacco, and Firearms strictly enforce the following limits. To receive the label 1984, Cabernet Sauvignon, Sonoma County, at least 95% of the contents must be of 1984 vintage, at least 75% of the contents must be Cabernet Sauvignon, and at least 85% must be from Sonoma County. How small might be the fraction of the wine in the bottle that is of 1984 vintage *and* of the Cabernet Sauvignon variety *and* from grapes grown in Sonoma County?
12. Rogers Foods of Turlock, California (see Rosenthal and Riefel (1994)) is a producer of high quality dried foods, such as dried onions, garlic, etc. It has regularly received "Supplier of the Year" awards from its customers, retail packaged food manufacturers such as Pillsbury. A reason for Rogers' quality reputation is it tries to supply product to its customers with quality characteristics that closely match customer specifications. This is difficult to do because Rogers does not have complete control over its input. Each food is harvested once per year from a variety of farms, one "lot" per farm. The quality of the crop from each farm is somewhat of a random variable. At harvest time, the crop is dried and each lot placed in the warehouse. Orders throughout the year are then filled from product in the warehouse.

Two of the main quality features of product are its density and its moisture content. Different customers may have different requirements for each quality attribute. If a product is too dense, then a jar that contains five ounces may appear only half full. If a product is not sufficiently dense, it may be impossible to get five ounces into a jar labelled as a five-ounce jar.

To illustrate the problem, suppose you have five lots of product with the following characteristics:

Lot	Fraction Moisture	Density	Kg. Available
1	0.03	0.80	1000
2	0.02	0.75	2500
3	0.04	0.60	3100
4	0.01	0.60	1500
5	0.02	0.65	4500

You currently have two prospective customers with the following requirements:

Customer	Fraction Moisture		Density		Max Kg. Desired	Selling Price per Kg.
	Min	Max	Min	Max		
<i>P</i>	0.035	0.045	0.70	0.75	3,000	\$5.25
<i>G</i>	0.01	0.03	0.60	0.65	15,000	\$4.25

What should you do?

13. The Lexus automobile gets 26 miles per gallon (mpg), the Corolla gets 31 mpg, and the Tercel gets 35 mpg. Let L , C , and T represent the number of automobiles of each type in some fleet. Let F represent the total number in the fleet. We require, in some sense, the mpg of the fleet to be at least 32 mpg. Fleet mpg is measured by (total miles driven by the fleet)/(total gallons of fuel consumed by fleet).
 - a) Suppose the sense in which mpg is measured is each auto is given one gallon of fuel, then driven until the fuel is exhausted. Write appropriate constraints to enforce the 32 mpg requirement.
 - b) Suppose the sense in which mpg is measured is each auto is driven one mile and then stopped. Write appropriate constraints to enforce the 32 mpg requirement.
14. In the financial industry, one is often concerned with the “duration” of one’s portfolio of various financial instruments. The duration of a portfolio is simply the weighted average of the duration of the instruments in the portfolio, where the weight is simply the number of dollars invested in the instrument. Suppose the Second National Bank is considering revising its portfolio and has denoted by X_1 , X_2 , and X_3 , the number of dollars invested (in millions) in each of three different instruments. The durations of the three instruments are respectively: 2 years, 4 years, and 5 years. The following constraint appeared in their planning model:

$$+ X_1 - X_2 - 2 \times X_3 \geq 0$$

In words, this constraint is:

- a) duration of the portfolio must be at most 10 years;
 - b) duration of the portfolio must be at least 3 years;
 - c) duration of the portfolio must be at least 2 years;
 - d) duration of the portfolio must be at most 3 years;
 - e) none of the above.
15. You are manager of a team of ditch diggers, each member of the team is characterized by a productivity measure with units of cubic feet per hour. An average productivity measure for the entire team should be based on which of the following:
- a) the arithmetic mean;
 - b) the geometric mean;
 - c) the harmonic mean.
16. Generic Foods has three different batches of cashews in its warehouse. The percentage moisture content for batches 1, 2, and 3 respectively are 8%, 11%, and 13%. In blending a batch of cashews for a particular customer, the following constraint appeared:

$$+ 2 \times X1 - X2 - 3 \times X3 \geq 0$$

In words, this constraint is:

- a) percent moisture must be at most 10%;
- b) percent moisture must be at least 3%;
- c) percent moisture must be at least 10%;
- d) percent moisture must be at most 2%;
- e) none of the above.

17. The Beanbody Company buys various types of raw coal on the open market and then pulverizes the coal and mixes it to satisfy customer specifications. Last week Beanbody bought 1500 tons of type *M* coal for \$78 per ton that was intended for an order that was canceled at the last minute. Beanbody had to pay an additional \$1 per ton to have the coal shipped to its processing facility. Beanbody has no other coal in stock. Type *M* coal has a BTU content of 13,000 BTU per ton. This week type *M* coal can be bought (or sold) on the open market for \$74 per ton. Type *W* coal, which has a BTU content of 10,000 BTU/ton, can be bought this week for \$68 per ton. Type *K* coal, which has a BTU content of 12,000 BTU/ton, can be bought this week for \$71 per ton. All require an additional \$1/ton to be shipped into Beanbody's facility. In fact, Beanbody occasionally sells raw coal on the open market and then Beanbody also has to pay \$1/ton outbound shipping. Beanbody expects coal prices to continue to drop next week. Right now Beanbody has an order for 2700 tons of pulverized product having a BTU content of at least 11,000 BTU per ton. Clearly, some additional coal must be bought. The president of Beanbody sketched out the following incomplete model for deciding how much of what coal to purchase to just satisfy this order.;

```

MODEL:
! MH = tons of on-hand type M coal used;
! MP = tons of type M coal purchased;
! WP = tons of type W coal purchased;
! KP = tons of type K coal purchased;

MIN = ___ * MH + ___ * MP + ___ * WP + ___ * KP;
MH + MP + WP + KP = 2700;
MH <= 1500;
2000 * MH + _____ * MP - 1000 * WP + _____ * KP >= 0;
END

```

What numbers would you place in the _____ places?

18. A local high school is considering using an outside supplier to provide meals. The big question is: How much will it cost to provide a nutritious meal to a student? Exhibit A reproduces the recommended daily minima for an adult as recommended by the noted dietitian, George Stigler (1945). Because our high school need provide only one meal per day, albeit the main one, it should be sufficient for our meal to satisfy one-half of the minimum daily requirements.

With regard to nutritive content of foods, Exhibit B displays the nutritional content of various foods available from one of the prospective vendors recommended by a student committee at the high school. See Bosch (1993) for a comprehensive discussion of these data.

For preliminary analysis, it is adequate to consider only calories, protein, calcium, iron, vitamins A, B1, and B2.

- Using only the candidate foods and prices in Exhibit B, and allowing fractional portions, what is the minimum cost needed to give a satisfactory meal at our high school?
- Suppose we require only integer portions be served in a meal (e.g., .75 of a Big Mac is not allowed). How is the cost per meal affected?
- Suppose in addition to (b), for meal simplicity, we put a limit of at most three food items from Exhibit B in a meal. For example, a meal of hamburger, fries, chicken McNuggets, and a garden salad has one too many items. How is the cost per meal affected?
- Suppose instead of (c), we require at most one unit per serving of a particular food type be used. How is the cost per meal affected?

- e) Suppose we modify (a) with the condition that the number of grams of fat in the meal must be less-than-or-equal-to $1/20$ th of the total calories in the meal. How is the cost per meal affected?
- f) How is the answer to (a) affected if you use current prices from your neighborhood McDonald's? For reference, Stigler claimed to be able to feed an adult in 1944 for \$59.88 for a full year.

Exhibit A

Nutrient	Allowance
Calories	3,000 calories
Protein	70 grams
Calcium	.8 grams
Iron	12 milligrams
Vitamin A	5,000 International Units
Thiamine (B1)	1.8 milligrams
Riboflavin (B2 or G)	2.7 milligrams
Niacin (Nicotinic Acid)	18 milligrams
Ascorbic Acid (C)	75 milligrams

Exhibit B

Menu Item	Price	Cal.	Protein	Fat	Sodium	Vit A	Vit C	Vit B1	Vit B2	Niacin	Calcium	Iron
Hamburger	0.59	255	12	9	490	4	4	20	10	20	10	15
McLean Deluxe	1.79	320	22	10	670	10	10	25	20	35	15	20
Big Mac	1.65	500	25	26	890	6	2	30	25	35	25	20
Small Fr. Fries	0.68	220	3	12	110	0	15	10	0	10	0	2
Ch. McNuggets	1.56	270	20	15	580	0	0	8	8	40	0	6
Chef Salad	2.69	170	17	9	400	100	35	20	15	20	15	8
Garden Salad	1.96	50	4	2	70	90	35	6	6	2	4	8
Egg McMuffin	1.36	280	18	11	710	10	0	30	20	20	25	15
Wheaties	1.09	90	2	1	220	20	20	20	20	20	2	20
Van. Cone	0.63	105	4	1	80	2	0	2	10	2	10	0
Milk	0.56	110	9	2	130	10	4	8	30	0	30	0
Orange Juice	0.88	80	1	0	0	0	120	10	0	0	0	0
Grapefruit Juice	0.68	80	1	0	0	0	100	4	2	2	0	0
Apple Juice	0.68	90	0	0	5	0	2	2	0	0	0	4

19. Your firm has just developed two new ingredients code named A and B . They seem to have great potential in the automotive aftermarket. These ingredients are blended in various combinations to produce a variety of products. For these products (and for the ingredients themselves), there are three qualities of interest: 1) opacity, 2) friction coefficient, and 3) adhesiveness. The research lab has provided the following table describing the qualities of various combinations of A and B :

Combination	Fraction of		Quality of this Combination		
	A	B	Opacity	Friction coef.	Adhesiveness
1	0.00	1.00	10.0	400.0	.100
2	0.50	0.50	25.0	480.0	.430
3	.75	.25	32.5	533.3	.522
4	1.00	0.00	40.0	600.0	.600

For example, the opacity of B by itself is 10, while the friction coefficient of A by itself is 600.

- For which qualities do the two ingredients appear to interact in a linear fashion?
 - You wish to prepare a product that, among other considerations, has opacity of at least 17, a friction coefficient of at least 430, and adhesiveness of no more than .35. Denote by T , A , and B the amount of total product produced, amount of A used, and the amount of B used. Write the constraints relating T , A , and B to achieve these qualities.
20. Indiana Flange Inc. produces a wide variety of formed steel products it ships to customers all over the country. It uses several different shipping companies to ship these products. The products are shipped in standard size boxes. A shipping company has typically two constraints it has to worry about in assembling a load: a weight constraint and a volume constraint. One of the shippers, Amarillo Freight, handles this issue by putting a density constraint (kilograms/cubic meter) on all shipments it receives from Indiana Flange. If the shipment has a density greater than a certain threshold (110 kg/m^3), Amarillo imposes a surcharge. Currently, Indiana Flange wants to ship the following products to Los Angeles:

Product	Long Tons	Density
A	100	130
B	85	95

Note, there are 1000 kilograms per long ton.

Let AY and BY be the number of tons shipped via Amarillo Freight. Although the densities of products A and B do not change from week to week, the number of tons Indiana Flange needs to ship varies considerably from week to week. Indiana Flange does not want to incur the surcharge. Write a constraint enforcing the Amarillo density constraint that is general (i.e., need not be changed from week to week).

21. The growth of the World Wide Web has dramatically increased the demand for glass fiber optic cable, the main medium used for high capacity data transfer. A major measure of quality of an optical fiber is its transmissivity, the fraction of the light transmitted into the fiber that is emitted out the other end. Suppose you are a fiber optics vendor, with an order in hand for a single 10 km strand of optic fiber, with a required transmissivity of at least .9875 per kilometer at a certain specified wave length of light. You have in stock two types of optic fiber, a) a very expensive one with a transmissivity of .992, and b) a relatively cheap one with a transmissivity of .982. It has

occurred to you that you could sell more of your fiber optic production if you could "blend" some of the lower quality fiber with the high quality. This is in fact possible by splicing a segment of one to the other, e.g., 8 km of the .992 fiber spliced onto 2 km of the .982 fiber. Suppose these splices are very high quality, i.e., no transmission loss across the splice. Let XH and XL be the length of high quality and low quality fiber you propose to use for the above request for 10 km of fiber. Write a constraint to ensure satisfaction of the customer's transmissivity requirement.